CPE 233 Hardware Assignment 6

Assembling the MCU

Report by:
Ethan Vosburg (evosburg@calpoly.edu)

March 9, 2024

Table of Contents

1 Project Description

In this project the central component of a functional microcontroller (MCU) is constructed by integrating various OTTER modules developed in previous assignments. While the resulting MCU will not represent the complete OTTER system, it will serve as a foundational unit for future expansions. This assignment will build the foundation for the rest of the hardware assignments as the OTTER starts to take shape and we can begin to implement it on the Basys3 board.

1.1 Control Unit Decoder

The Control Unit Decoder is a combinational logic component responsible for generating control signals based on the current instruction. Unlike the Control Unit FSM, the signals generated by the decoder are less time-sensitive, allowing for a simpler, combinational design. The decoder takes the instruction opcode as input and outputs control signals to various components of the MCU, such as the ALU, registers, and memory units. The primary function of the decoder is to interpret the instruction and set the appropriate control signals to execute the operation specified by the instruction.

1.2 Control Unit FSM

The Control Unit FSM is responsible for controlling the timing-critical aspects of the MCU's operation. It operates in three main states: FETCH, EXEC, and WRITEBACK. The FSM ensures that instructions are fetched, executed, and the results are written back in a controlled manner, particularly for operations involving memory access.

1.3 OTTER MCU

The OTTER MCU is a microcontroller unit composed of various modules, including the Control Unit (with its FSM and Decoder), ALU, registers, and memory units. It is designed to execute a set of instructions, manipulating data and performing operations as specified by the program.

2 Structural Design

2.1 OTTER MCU



Figure 1: OTTER MCU Eleborated Design

3 Synthesis Warnings

3.1 MCU Synthesis Warnings



Figure 2: Synthesis Warnings for MCU

The warnings referenced in Figure 2 are not important as there is hardware that will still need to be implemented in future assignments. This means there are various wires that do not have a driver or that do not lead to a defined location. This is fine and in this case does not prevent the design from working. Vivado is simply letting you know that these lines will not be implemented in to the design at this point.

4 Verification



Figure 3: OTTER MCU Verification

Address	Code	Basic	Source
0x00000000	0xaa0552b7	lui x5,0x000aa055	2: lui x5, 0xAA055
0x00000004	0x76528413	addi x8,x5,0x00000765	3: addi x8, x5, 0x765
0x00000008	0x00341513	slli x10,x8,3	4: slli x10, x8, 3
0x0000000c	0x0082a633	slt x12,x5,x8	5: slt x12, x5, x8
0x00000010	0x00a446b3	xor x13,x8,x10	6: xor x13, x8, x10
0x00000014	0xfe0006e3	beq x0,x0,0xfffffffec	7: beq x0, x0, main

Name	Number	Value
zero	0	0x00000000
ra	1	0x00000000
sp	2	0x00000000
gp	3	0x00000000
tp	4	0x00000000
t0	5	0xaa055000
t1	6	0x00000000
t2	7	0x00000000
s0	8	0xaa055765
s1	9	0x00000000
a0	10	0x502abb28
a1	11	0x00000000
a2	12	0x00000001
a3	13	0xfa2fec4d

Figure 4: OTTER MCU RARS Simulation

Looking at the simulation we have the outputs that we would expect referencing figure 4 you can compare the hex values for the registers and see that they all match up. They registers have an undefined value before they are written to which is expected. Also There is a longer INIT state becuase of the reset command in the beginning of the file.

5 Source Code

5.1 Branch Address Generator Source Code

Listing 1: Control Unit FSM

```
`timescale 1ns / 1ps
//////////////////////////////////////////////////////////////////////////////////
// Company: Cal Poly SLO
// Engineer: Ethan Vosburg
//
// Create Date: 02/21/2024 03:42:56 PM
// Module Name: ControlUnitFSM
// Project Name: Risc-V MCU
// Target Devices: Basys3
// Description: Control Unit FSM for the Risc-V MCU
//
// Revision:
// Revision 0.01 - File Created
//
//////////////////////////////////////////////////////////////////////////////////


module ControlUnitFSM(
    // Inputs
    input CLK,                              // Clock
    input RST,                              // Reset
    input INTR,                             // Interrupt
    input [6:0] opcode,                     // Opcode
    input [2:0] funct3,                     // Function 3

    // Outputs
    output logic PC_WE,                     // Program Counter Write Enable
    output logic RF_WE,                     // Register File Write Enable
    output logic memWE2,                    // Memory Write Enable 2
    output logic memRDEN1,                  // Memory Read Enable 1
    output logic memRDEN2,                 // Memory Read Enable 2
    output logic reset,                     // Reset
    output logic csr_WE,                    // Control and Status Register Write Enable
    output logic int_taken,                 // Interrupt
    output logic mret_exec                  // MRET Execution
);
    // Define the state type
    typedef enum { ST_INIT, ST_FETCH, ST_EXEC, ST_WB, ST_INTR} state_type;

    //  Define state and next state
    state_type state, next_state;

    // Transition State
    always_ff @(posedge CLK) begin
        if (RST) begin
            state <= ST_INIT;
        end else begin
            state <= next_state;
        end
    end


    // Define the state logic
    always_comb begin
        // Initialize all the outputs to zeros
        PC_WE = 1'b0;
        RF_WE = 1'b0;
        memWE2 = 1'b0;
        memRDEN1 = 1'b0;
        memRDEN2 = 1'b0;
```

```
61          reset = 1'b0;
62          csr_WE = 1'b0;
63          int_taken = 1'b0;
64          mret_exec = 1'b0;
65
66          case (state)
67              ST_INIT: begin
68                  reset = 1'b1;
69                  next_state = ST_FETCH;
70              end
71              ST_FETCH: begin
72                  memRDEN1 = 1'b1;
73                  next_state = ST_EXEC;
74              end
75              ST_EXEC: begin
76                  case (opcode)
77                  // J-Type Instruction
78                  7'b1101111: begin
79                      PC_WE = 1'b1;
80                      RF_WE = 1'b1;
81                      next_state = ST_FETCH;
82                  end
83                  // R-Type Instruction
84                  7'b0110011: begin
85                      PC_WE = 1'b1;
86                      RF_WE = 1'b1;
87                      next_state = ST_FETCH;
88                  end
89                  // I-Type Instruction
90                  7'b0010011: begin
91                      PC_WE = 1'b1;
92                      memRDEN1 = 1'b1;
93                      RF_WE = 1'b1;
94                      next_state = ST_FETCH;
95                  end
96                  // JALR Instruction
97                  7'b1100111: begin
98                      PC_WE = 1'b1;
99                      memRDEN1 = 1'b1;
100                     RF_WE = 1'b1;
101                     next_state = ST_FETCH;
102                 end
103                 // Loading Instructions
104                 7'b0000011: begin
105                     PC_WE = 1;
106                     memRDEN1 = 1'b1;
107                     memRDEN2 = 1;
108                     RF_WE = 1'b1;
109                     next_state = ST_WB;
110                 end
111                 // B-Type Instruction
112                 7'b1100011: begin
113                     PC_WE = 1'b1;
114                     next_state = ST_FETCH;
115                 end
116                 // U-Type Instruction
117                 7'b0110111: begin
118                     PC_WE = 1'b1;
119                     RF_WE = 1'b1;
120                     memRDEN1 = 1'b1;
121                     next_state = ST_FETCH;
122                 end
123                 // AUIPC Instruction
124                 7'b0010111: begin
125                     PC_WE = 1'b1;
126                     RF_WE = 1'b1;
127                     memRDEN1 = 1'b1;
128                     next_state = ST_FETCH;
129                 end
130                 // S-Type Instructions
```

```
131                 7'b0100011: begin
132                     PC_WE = 1'b1;
133                     memRDEN1 = 1'b1;
134                     memWE2 = 1;
135                     next_state = ST_FETCH;
136                 end
137 
138                 default: next_state = ST_INIT; // Should never happen
139                 endcase
140             end
141             ST_WB: begin
142                 RF_WE = 1'b1;
143                 PC_WE = 0;
144                 next_state = ST_FETCH;
145             end
146             ST_INTR: begin
147                 PC_WE = 1;
148                 int_taken = 1;
149             end
150             default: next_state = ST_INIT; // Should never happen
151         endcase
152     end
153 endmodule
```

Listing 2: Control Unit Decoder

```
'timescale 1ns / 1ps
//////////////////////////////////////////////////////////////////////////////////
// Company: Cal Poly SLO
// Engineer: Ethan Vosburg
//
// Create Date: 02/21/2024 03:47:15 PM
// Module Name: ControlUnitDecoder
// Project Name: Otter MCU
// Target Devices: Basys3
// Description: Control Unit Decoder for controlling the ALU
//
// Revision:
// Revision 0.01 - File Created
//
//////////////////////////////////////////////////////////////////////////////////


module ControlUnitDecoder(
    // Inputs
    input br_eq,                            // Branch Equal
    input br_lt,                            // Branch Less Than
    input br_ltu,                           // Branch Less Than Unsigned
    input [6:0] funct7,                     // Function 7
    input [6:0] opcode,                     // Opcode
    input [2:0] funct3,                     // Function 3

    // Outputs
    output logic [3:0] ALU_FUN,             // ALU Function
    output logic [1:0] srcA_SEL,            // Source A Select
    output logic [2:0] srcB_SEL,            // Source B Select
    output logic [2:0] PC_SEL,              // Program Counter Select
    output logic [1:0] RF_SEL               // Register File Select
    );

    always_comb begin
        // Initialize all the outputs to zeros
        ALU_FUN = 4'b0000;
        srcA_SEL = 2'b0;
        srcB_SEL = 3'b0;
        PC_SEL = 3'b0;
        RF_SEL = 2'b00;

        case (opcode)
        // R-Type Instruction
        7'b0110011: begin
            ALU_FUN = {funct7[5], funct3};
            srcA_SEL = 2'b0;
            srcB_SEL = 0;
            PC_SEL = 3'b0;
            RF_SEL = 2'b11;
            // Logic for ALU Select B
            // if ({funct7[5], funct3} == 4'b0010) srcB_SEL = 3'b1;
            // if ({funct7[5], funct3} == 4'b0100) srcB_SEL = 3'b0;
        end
        // I-Type Instruction
        7'b0010011: begin
            // Take care of special case with SRAI
            if (funct3 == 3'b101)
                ALU_FUN = {funct7[5], funct3};
            else
                ALU_FUN = {1'b0, funct3};

            // Configure selects
            srcA_SEL = 2'b0;
            srcB_SEL = 3'b1;
            RF_SEL = 2'b11;
            PC_SEL = 3'b0;
        end
        // JALR Instructions
```

```
        7'b1100111: begin
            PC_SEL = 3'b1;
            // ALU_FUN = {1'b0, funct3};
            // srcA_SEL = 2'b0;
            // srcB_SEL = 3'b1;
            RF_SEL = 2'b0;
        end
        // Loading Instructions
        7'b0000011: begin
            PC_SEL = 3'b0;
            RF_SEL = 2;
            srcA_SEL = 0;
            srcB_SEL = 1;
            ALU_FUN = 4'b0000;
        end
        // J-Type Instruction
        7'b1101111: begin
            PC_SEL = 3'b11;
            RF_SEL = 2'b0;
        end
        // B-Type Instruction
        7'b1100011: begin
            case(funct3)
                3'b000: begin
                    if (br_eq)
                        PC_SEL = 3'b10;
                    else
                        PC_SEL = 3'b0;
                end

                3'b101: begin
                    if (!br_lt)
                        PC_SEL = 3'b10;
                    else
                        PC_SEL = 3'b0;
                end

                3'b111: begin
                    if (!br_ltu)
                        PC_SEL = 3'b10;
                    else
                        PC_SEL = 3'b0;
                end

                3'b100: begin
                    if (br_lt)
                        PC_SEL = 3'b10;
                    else
                        PC_SEL = 3'b0;
                end

                3'b110: begin
                    if (br_ltu)
                        PC_SEL = 3'b10;
                    else
                        PC_SEL = 3'b0;
                end

                3'b001: begin
                    if (!br_eq)
                        PC_SEL = 3'b10;
                    else
                        PC_SEL = 3'b0;
                end

                default: PC_SEL = 3'b111;
            endcase
        end
        // U-Type Instruction
        7'b0110111: begin
```

```
            ALU_FUN = 4'b1001;
            srcA_SEL = 2'b1;
            PC_SEL = 3'b0;
            RF_SEL = 2'b11;
        end

        // AUIPC Command
        7'b0010111: begin
            ALU_FUN = 4'b0000;
            srcA_SEL = 2'b1;
            srcB_SEL = 3'b11;
            PC_SEL = 3'b0;
            RF_SEL = 2'b11;
        end
        // S-Type Instruction
        7'b0100011: begin
            ALU_FUN = 0;
            srcA_SEL = 0;
            srcB_SEL = 2;
            PC_SEL = 0;
        end
        // Interrupt functions
        7'b1110011: begin
            case(funct3)
                // CSRRW
                3'b001: begin
                    ALU_FUN = 4'b1001;
                    RF_SEL = 2'b01;
                end
                3'b010: begin
                    ALU_FUN = 4'b0110;
                    srcB_SEL = 3'b100;
                    RF_SEL = 2'b01;
                end
                3'b011:begin
                    ALU_FUN = 4'b0111;
                    srcA_SEL = 2'b10;
                    srcB_SEL = 3'b100;
                    RF_SEL = 2'b01;
                end

            endcase


        end

        default: begin
            // Should not be used
            ALU_FUN = 4'b0000;
            srcA_SEL = 2'b1;
            srcB_SEL = 3'b11;
            PC_SEL = 3'b0;
            RF_SEL = 2'b11;
        end
        endcase

    end


endmodule
```

Listing 3: Master MCU Linking all Modules Together

```
`timescale 1ns / 1ps
//////////////////////////////////////////////////////////////////////////////////
// Company: Cal Poly SLO
// Engineer: Ethan Vosburg
//
// Create Date: 02/21/2024 03:42:56 PM
// Module Name: MCU
// Project Name: Risc-V MCU
// Target Devices: Basys3
// Description: MCU linking all the submodules together
//
// Revision:
// Revision 0.01 - File Created
//
//////////////////////////////////////////////////////////////////////////////////


module MCU(
    // Inputs
    input CLK,                          // Clock
    input RST,                          // Reset
    input INTR,                         // Interrupt
    input [31:0] IO_BUS_IN,             // IO Bus In

    // Outputs
    output [31:0] IOBUS_OUT,            // IO Bus Out
    output [31:0] IOBUS_ADDR,           // IO Bus Address
    output IOBUS_WR                     // IO Bus Write/Read Data
    );

    // Internal Wires
    logic [31:0] pc_mux, pc_count, pc_count_inc, alu_result, rs1, rs2, ir, ir2,
            rf_mux, utype, itype, stype, jtype, btype, csr_rd, jalr, branch,
            jal,alu_srca_mux, alu_srcb_mux, mtvec, mepc;
    logic pc_we, rf_we, memwe2, memrden1, memrden2, reset, csr_we, int_taken,
            mret_exec;
    logic [3:0] alu_fun;
    logic [1:0] alu_srca_mux_select, rf_mux_select;
    logic [2:0] alu_srcb_mux_select, pc_mux_select;

    // Submodules
    // Assign statments for single lines
    assign IOBUS_OUT = rs2;
    assign IOBUS_ADDR = alu_result;
    assign pc_count_inc = pc_count + 4;

    // Linking all modules together
    ProgramCounter PC(
        .PC_RST(RST),
        .PC_WE(pc_we),
        .PC_DIN(pc_mux),
        .CLK(CLK),
        .PC_COUNT(pc_count)
    );

    ProgramCounterMux PCM(
        .PC_COUNT_INCD(pc_count_inc),
        .JALR(jalr),
        .BRANCH(branch),
        .JAL(jal),
        .MTVEC(mtvec),
        .MEPC(mepc),
        .MUX_SELECT(pc_mux_select),
        .PC_MUX_OUT(pc_mux)
    );

    OtterMemory MEM(
        .MEM_CLK(CLK),
        .MEM_RDEN1(memrden1),
```

```
        .MEM_RDEN2(memrden2),
        .MEM_WE2(memwe2),
        .MEM_ADDR1(pc_count[15:2]),
        .MEM_ADDR2(alu_result),
        .MEM_DIN2(rs2),
        .MEM_SIZE(ir[13:12]),
        .MEM_SIGN(ir[14]),
        .IO_IN(IO_BUS_IN),
        .IO_WR(IOBUS_WR),
        .MEM_DOUT1(ir),
        .MEM_DOUT2(ir2)
        );

    RegFile RF(
        .CLK(CLK),
        .ADR1(ir[19:15]),
        .ADR2(ir[24:20]),
        .WADR(ir[11:7]),
        .ENABLE(rf_we),
        .WDATA(rf_mux),
        .RS1(rs1),
        .RS2(rs2)
    );

    ImmdGen IMM(
        .instruction(ir),
        .uTypeImmd(utype),
        .iTypeImmd(itype),
        .sTypeImmd(stype),
        .jTypeImmd(jtype),
        .bTypeImmd(btype)
    );

    BranchAddressGen BAG(
        .programCounter(pc_count),
        .jTypeImmd(jtype),
        .bTypeImmd(btype),
        .iTypeImmd(itype),
        .sourceReg1(rs1),
        .jal(jal),
        .branch(branch),
        .jalr(jalr)
    );

    ALU ALU(
        .srcA(alu_srca_mux),
        .srcB(alu_srcb_mux),
        .operation(alu_fun),
        .result(alu_result)
    );

    BranchConditionGen BCG(
        .sourceReg1(rs1),
        .sourceReg2(rs2),
        .equal(br_eq),
        .isLess(br_lt),
        .isLessUnsigned(br_ltu)
    );

    ControlUnitDecoder CUD(
        .br_eq(br_eq),
        .br_lt(br_lt),
        .br_ltu(br_ltu),
        .funct7(ir[31:25]),
        .opcode(ir[6:0]),
        .funct3(ir[14:12]),
        .ALU_FUN(alu_fun),
        .srcA_SEL(alu_srca_mux_select),
        .srcB_SEL(alu_srcb_mux_select),
        .PC_SEL(pc_mux_select),
```

```
        .RF_SEL(rf_mux_select)
    );

    ControlUnitFSM CUF(
        .CLK(CLK),
        .RST(RST),
        .INTR(INTR),
        .opcode(ir[6:0]),
        .funct3(ir[14:12]),
        .PC_WE(pc_we),
        .RF_WE(rf_we),
        .memWE2(memwe2),
        .memRDEN1(memrden1),
        .memRDEN2(memrden2),
        .reset(reset),
        .csr_WE(csr_we),
        .int_taken(int_taken),
        .mret_exec(mret_exec)
    );

    RegFileMux RFM(
        .RF_SEL(rf_mux_select),
        .PC_COUNT_INC(pc_count_inc),
        .csr_RD(csr_rd),
        .MemoryDOUT2(ir2),
        .ALU_RESULT(alu_result),
        .muxOut(rf_mux)
    );

    ALUsrcAMux ALUsrcAM(
        .srcA_SEL(alu_srca_mux_select),
        .RS1(rs1),
        .uTypeImmd(utype),
        .notRS1(~rs1),
        .muxOut(alu_srca_mux)
    );

    ALUsrcBMux ALUsrcBM(
        .srcB_SEL(alu_srcb_mux_select),
        .RS2(rs2),
        .iTypeImmd(itype),
        .sTypeImmd(stype),
        .PC_COUNT(pc_count),
        .csr_RD(csr_rd),
        .muxOut(alu_srcb_mux)
    );

endmodule
```

6 Conclusion

In this project, the MCU was sussesfully constructed and tested with a limited set of code. The ground work for future hardware compnents was made tested to be working. This is the culmination of all the modules that have been made to this point and represents the first time that the compnents have all worked together. From here more moduels can be added to add more funcitonality. All code for this assignment can be found here.